PRÁCTICA 7: Ecuaciones Diferenciales.

Ecuaciones Diferenciales

- Para resolver simbólicamente, con *wxMaxima*, ecuaciones diferenciales ordinarias de primer y segundo orden utilizaremos la instrucción **ode2.**Esta orden está en las paletas en el apartado **Ecuaciones,** ir a, **Resolver EDO.**
- Para expresar las ecuaciones se hace uso de **'diff** , para representar la derivada.
- Para definir funciones utilizaremos la orden **define.**

Ejemplo 1

Hallar la solución general de la ecuación diferencial :

$$x^3(y-1)\frac{dy}{dx}+(x-1)y^3=0$$

Solución con wxMaxima:

```
(%i1)   ec:(x-1)*y^3+(y-1)*x^3*'diff(y,x,1)=0;
```

(ec) $x^3\left(y-1\right)\left(\frac{d}{d\,x}\,y\right)+\left(x-1\right)y^3=0$

```
(%i2)   sg:ode2(ec,y,x);
```

(sg) $\frac{2\,y-1}{2\,y^2}=\%c-\frac{2\,x-1}{2\,x^2}$

Podemos observar que aparece en la solución general la expresión *%c*, que es la constante genérica, que toma todos los valores reales.
Como sabemos, si nos dan condiciones iniciales, podemos hallar la solución particular correspondiente a un cierto valor de la constante *%c.*

- Esto lo haremos, con *wxMaxima*, utilizando la instrucción **ic1.**Esta orden está en las paletas en el apartado **Ecuaciones,** ir a, **Problema de valor inicial (1).**

Ejemplo 2

Hallar la solución particular que verifica las condiciones iniciales:

$$\begin{cases} x=2 \\ y=\ \ 3 \end{cases}$$

Solución con wxMaxima:

```
(%i3)   sp:ic1(sg,x=2,y=3);
```

(sp) $\frac{2\,y-1}{2\,y^2}=\frac{47\,x^2-72\,x+36}{72\,x^2}$

Podemos representar gráficamente, dicha solución particular.

```
(%i5)   wxdraw2d(color=red,implicit(sp,x,0,7,y,0,7));
```

(%t5)



(%o5)

Ejemplo 3

Dada la ecuación diferencial :

$$\frac{dy}{dx} - y + \sqrt{y} = 0$$

a) Hallar la solución general.
b) Hallar , en forma explícita, la solución particular que verifica las condiciones iniciales:

$$\begin{cases} x = 0 \\ y = 4 \end{cases}$$

c) Representar gráficamente dicha solución particular

Solución con wxMaxima:

(%i6) `ec:'diff(y,x)-y+sqrt(y)=0;`

(ec) $\frac{d}{d\,x}\,y - y + \sqrt{y} = 0$

(%i7) `sg:ode2(ec,y,x);`

(sg) $2\,\log\left(\sqrt{y}-1\right) = x + \%c$

(%i8) `sp:ic1(sg,x=0,y=4);`

(sp) $2\,\log\left(\sqrt{y}-1\right) = x$

```
-->     /* Despejamos la y */
(%i9)   spe:solve(sp,y);
```

(spe) $[y = \%e^{x} + 2\,\%e^{x/2} + 1]$

```
-->     /* Definimos la función en forma explícita */
(%i10)  solpe:y,spe;
```

(solpe) $\%e^{x} + 2\,\%e^{x/2} + 1$

```
(%i11)  define(y(x),solpe);
```

(%o11) $y(x) := \%e^{x} + 2\,\%e^{x/2} + 1$

```
(%i12)  wxdraw2d(color=red,explicit(y(x),x,-5,5));
```

(%t12)



(%o12)

Veamos un ejemplo de ecuación diferencial de segundo orden.

Ejemplo 4

Hallar la solución general de la ecuación diferencial :

$$\frac{d^2y}{dx^2} - \frac{dy}{dx} + x = 0$$

Solución con wxMaxima:

```
(%i13)  ec:'diff(y,x,2)-'diff(y,x)+x=0;
```

(ec) $\frac{d^2}{d\,x^2}\,y - \frac{d}{d\,x}\,y + x = 0$

```
(%i14)    sg:ode2(ec,y,x);
```

(sg) $$y=\%k1\ \%e^{x}+\frac{x^{2}+2\,x+2}{2}+\%k2$$

Ahora aparecen en la solución general las expresiónes *%k1* y , *%k2* que son las constantes genérica, que toman todos los valores reales.
Si nos dan condiciones iniciales, podemos hallar la solución particular correspondiente a uno ciertos valores de las constantes *%k1* y , *%k2*.

- Esto lo haremos, con *wxMaxima*, utilizando la instrucción **ic2.**Esta orden está en las paletas en el apartado **Ecuaciones,** ir a, **Problema de valor inicial (2).**

Hallar la solución particular que verifica las condiciones iniciales:

$$\begin{cases} x=1 \\ y=-1 \\ \frac{dy}{dx}=2 \end{cases}$$

```
(%i16)    sp:ic2(sg,x=1,y=-1,diff(y,x)=2);
```

(sp) $$y=\frac{x^{2}+2\,x+2}{2}-\frac{7}{2}$$

En vez de condiciones iniciales, podemos imponer condiciones de contorno.

- Esto lo haremos, con *wxMaxima*, utilizando la instrucción **bc2.**Esta orden está en las paletas en el apartado **Ecuaciones,** ir a, **Problema de contorno.**

Hallar la solución particular que verifica las condiciones ide contorno:

$$\begin{cases} x=1 \\ y=3 \end{cases} ; \begin{cases} x=2 \\ y=\frac{5}{3} \end{cases}$$

```
(%i20)    bc2(sg,x=1,y=3,x=2,y=5/3);
```

(%o20) $$y=-\frac{23\ \%e^{x}}{6\ \%e^{2}-6\ \%e}+\frac{x^{2}+2\,x+2}{2}+\frac{3\ \%e+20}{6\ \%e-6}$$

Ejemplo 5

La ley de Newton para el enfriamiento afirma que la rapidez con que un cuerpo se enfria es proporcional a la diferencia entre la temperatura del cuerpo y la del medio que lo rodea.
Un cuerpo con una temperatura de 80^{o}F se coloca en el instante t=0, en un medio cuya temperatura se mantiene a 50^{o}F, y al cabo de cinco minutos, el cuerpo se ha enfriado hasta los 70^{o}F.
Hállese la temperatura del cuerpo al cabo de diez minutos

Solución con wxMaxima:

```
-->       /* Planteamos la ecuación */
```

```
(%i22) ecu:'diff(x,t)=k*(x-50);
```

(ecu) $\frac{d}{d\,t}\,x=k\,(x-50)$

```
-->    /* Solución general */
(%i23) sg:ode2(ecu,x,t);
```

(sg) $x=\left(50\ \%e^{-k\,t}+\%c\right)\ \%e^{k\,t}$

```
-->    /* Solución particular */
(%i24) sp:ic1(sg,t=0,x=80);
```

(sp) $x=30\ \%e^{k\,t}+50$

```
(%i25) spe:x,sp;
```

(spe) $30\ \%e^{k\,t}+50$

```
-->    /* Definimos la función */
(%i26) define(x(k,t),spe);
```

(%o26) $\mathrm{x}(k,t):=30\ \%e^{k\,t}+50$

```
-->    /* Hallamos k */

(%i27) raiz:solve(x(k,5)=70,k);
```

(raiz) $\left[k=\log\left(\frac{2^{1/5}\ \%e^{\frac{2\,\%i\,\pi}{5}}}{3^{1/5}}\right),k=\log\left(\frac{2^{1/5}\ \%e^{\frac{4\,\%i\,\pi}{5}}}{3^{1/5}}\right),k=\frac{5\log\left(\frac{2^{1/5}}{3^{1/5}}\right)-4\,\%i\,\pi}{5},k=\frac{5\log\left(\frac{2^{1/5}}{3^{1/5}}\right)-2\,\%i\,\pi}{5},k=\log\left(\frac{2^{1/5}}{3^{1/5}}\right)\right]$

```
-->    /* Elegimos la raíz correcta */
(%i28) h:k,raiz[5];
```

(h) $\log\left(\frac{2^{1/5}}{3^{1/5}}\right)$

```
(%i29) x(h,10),numer;
(%o29) 63.33333333333335
```

```
(%i30) wxdraw2d(color=red,explicit(x(h,t),t,0,15));
```

(%t30)



(%o30)

EJERCICIOS PROPUESTOS

1.

La población de una ciudad aumenta con un coeficiente de variación porcional al número de habitantes en cada instante. Si la población era de 30.000 habitantes hace 10 años y de 35.000 en la actualidad, ¿cuál será su población dentro de 10 años?.

Solución: 40833. 33333333331

2.

La ecuación que rige el movimiento de un cuerpo que se desliza por un plano inclinado es:

$$ma = mgsen\alpha - \mu mg\cos\alpha - R,$$

siendo m la masa, a la aceleración, g la gravedad, α el ángulo del plano inclinado, R la resistencia del aire y μ el coeficiente de rozamiento. Se abandona un objeto de 6 kilogramos de masa desde la parte superior de un plano inclinado un ángulo de 30^o respecto a la horizontal y se supone que, numéricamente, la resistencia del aire (en newtons) es tres veces la velocidad (en metros por segundo) y que el coeficiente de rozamiento es $1/5$. Hállese la velocidad en función del tiempo, $v(t)$, y represéntese gráficamente. ¿cuál será la velocidad del objeto al cabo de cuatro segundos de ser abadonado?.

Solución: 5. 538333511226371

3.

Sabiendo que en un circuito RLC eléctrico en serie se cumple:

$$L\frac{d^2I}{dt^2}+R\frac{dI}{dt}+\frac{1}{C}I=\frac{dE}{dt}$$

siendo L la inductancia, I la intensidad, R la resistencia, C la capacitancia y E la fuerza eleztrómotriz y que los valores de los distintos elementos son $L=0.001$ henrios, $R=0.02$ ohmios, $C=2$ faradios y $E(t)=sen100t$ voltios y teniendo en cuenta que la intensidad I y su derivada $\frac{dI}{dt}$ son cero, en el instante inicial, hállese la intensidad en función del tiempo, $I(t)$, y represéntese gráficamente. ¿cuál será la intensidad al cabo de dos segundos ?.

Solución: -6.763795785800259

4.

Un depósito contiene inicialmente 100 litros de una solución en la que hay disueltos 20 grs de sal. En el instante t=0 comienza a entrar en el depósito una solución que contiene 3 gramos de sal por litro, a razón de 4 litros por minuto. Se conserva uniforme la mezcla agitándola y fluye simultáneamente al exterior, a la misma velocidad. Hállese la cantidad de sal en función del tiempo $x(t)$ y represéntese gráficamente. ¿cuánta sal habrá en el deposito al cabo de 10 minutos?

Solución: 112.310387110021

5.

a) Hallar la solución general de la ecuación diferencial :

$$\frac{d^2y}{dx^2}+\frac{dy}{dx}=e^x$$

b) Hallar y representar gráficamente la solución particular que verifica las condiciones iniciales:

$$\begin{cases} x=0 \\ y=0 \\ \frac{dy}{dx}=1 \end{cases}$$

b) Hallar y representar gráficamente la solución particular que verifica las condiciones de contorno:

$$\begin{cases} x=0 \\ y=1 \end{cases} ; \begin{cases} x=1 \\ y=0 \end{cases}$$

Solución: $a)\ y=\frac{e^x}{2}+k2*e^{-x}+k1\ b)\ y=\frac{e^x}{2}-\frac{e^{-x}}{2}\ c)\ y=\frac{e^x}{2}+\frac{(e^2+e)*e^{-x}}{2*e-2}-\frac{e^2+1}{2*e-2}$

6.

Un cuerpo de 4 kilogramos de masa cae, partiendo del reposo desde una gran altura. Conforme cae, actúa sobre él la resistencia del aire (en Newtons) que es igual a 28 ceces la velocidad (en m/seg.). Cáculense la velocidad v(t) y la distancia recorrida al cabo de 10 segundos. Represéntense las gráficas de la distancia y la velocidad en el intervalo $[0,3]$.

Solución: $a)\ v(10)=1.4\ b)\ x(10)=13.8$